

NAG Library Routine Document

G05PGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

G05PGF generates a given number of terms of an exponential GARCH(p, q) process (see Engle and Ng (1993)).

2 Specification

SUBROUTINE G05PGF (DIST, NUM, IP, IQ, THETA, DF, HT, ET, FCALL, R, LR, &
STATE, IFAIL)

INTEGER NUM, IP, IQ, DF, LR, STATE(*), IFAIL
REAL (KIND=nag_wp) THETA(2*IQ+IP+1), HT(NUM), ET(NUM), R(LR)
LOGICAL FCALL
CHARACTER(1) DIST

3 Description

An exponential GARCH(p, q) process is represented by:

$$\ln(h_t) = \alpha_0 + \sum_{i=1}^q \alpha_i z_{t-i} + \sum_{i=1}^q \phi_i (|z_{t-i}| - E[|z_{t-i}|]) + \sum_{j=1}^p \beta_j \ln(h_{t-j}), \quad t = 1, 2, \dots, T;$$

where $z_t = \frac{\epsilon_t}{\sqrt{h_t}}$, $E[|z_{t-i}|]$ denotes the expected value of $|z_{t-i}|$, and $\epsilon_t | \psi_{t-1} = N(0, h_t)$ or $\epsilon_t | \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t -distribution with df degrees of freedom and variance h_t , T is the number of observations in the sequence, ϵ_t is the observed value of the GARCH(p, q) process at time t , h_t is the conditional variance at time t , and ψ_t the set of all information up to time t .

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05PGF.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and testing the impact of news on volatility *Journal of Finance* **48** 1749–1777

Glosten L, Jagannathan R and Runkle D (1993) Relationship between the expected value and the volatility of nominal excess return on stocks *Journal of Finance* **48** 1779–1801

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Arguments

- 1: DIST – CHARACTER(1) *Input*
On entry: the type of distribution to use for ϵ_t .
 DIST = 'N'
 A Normal distribution is used.
 DIST = 'T'
 A Student's t -distribution is used.
Constraint: DIST = 'N' or 'T'.
- 2: NUM – INTEGER *Input*
On entry: T , the number of terms in the sequence.
Constraint: NUM ≥ 0 .
- 3: IP – INTEGER *Input*
On entry: the number of coefficients, β_i , for $i = 1, 2, \dots, p$.
Constraint: IP ≥ 0 .
- 4: IQ – INTEGER *Input*
On entry: the number of coefficients, α_i , for $i = 1, 2, \dots, q$.
Constraint: IQ ≥ 1 .
- 5: THETA($2 \times$ IQ + IP + 1) – REAL (KIND=nag_wp) array *Input*
On entry: the initial parameter estimates for the vector θ . The first element must contain the coefficient α_0 and the next IQ elements must contain the autoregressive coefficients α_i , for $i = 1, 2, \dots, q$. The next IQ elements must contain the coefficients ϕ_i , for $i = 1, 2, \dots, q$. The next IP elements must contain the moving average coefficients β_j , for $j = 1, 2, \dots, p$.
Constraints:
- $$\sum_{i=1}^p \beta_i \neq 1.0;$$
- $$\frac{\alpha_0}{1 - \sum_{i=1}^p \beta_i} \leq -\log(\text{X02AMF}).$$
- 6: DF – INTEGER *Input*
On entry: the number of degrees of freedom for the Student's t -distribution.
 If DIST = 'N', DF is not referenced.
Constraint: if DIST = 'T', DF > 2 .
- 7: HT(NUM) – REAL (KIND=nag_wp) array *Output*
On exit: the conditional variances h_t , for $t = 1, 2, \dots, T$, for the GARCH(p, q) sequence.
- 8: ET(NUM) – REAL (KIND=nag_wp) array *Output*
On exit: the observations ϵ_t , for $t = 1, 2, \dots, T$, for the GARCH(p, q) sequence.
- 9: FCALL – LOGICAL *Input*
On entry: if FCALL = .TRUE., a new sequence is to be generated, otherwise a given sequence is to be continued using the information in R.

- 10: R(LR) – REAL (KIND=nag_wp) array *Input/Output*
On entry: the array contains information required to continue a sequence if FCALL = .FALSE..
On exit: contains information that can be used in a subsequent call of G05PGF, with FCALL = .FALSE..
- 11: LR – INTEGER *Input*
On entry: the dimension of the array R as declared in the (sub)program from which G05PGF is called.
Constraint: $LR \geq 2 \times (IP + 2 \times IQ + 2)$.
- 12: STATE(*) – INTEGER array *Communication Array*
Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.
On entry: contains information on the selected base generator and its current state.
On exit: contains updated information on the state of the generator.
- 13: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
On exit: IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, DIST is not valid: DIST = $\langle value \rangle$.

IFAIL = 2

On entry, NUM = $\langle value \rangle$.
 Constraint: NUM \geq 0.

IFAIL = 3

On entry, IP = $\langle value \rangle$.
 Constraint: IP \geq 0.

IFAIL = 4

On entry, IQ = $\langle value \rangle$.
 Constraint: IQ \geq 1.

IFAIL = 6

On entry, $DF = \langle value \rangle$.
Constraint: $DF \geq 3$.

IFAIL = 10

IP or IQ is not the same as when R was set up in a previous call.
Previous value of $IP = \langle value \rangle$ and $IP = \langle value \rangle$.
Previous value of $IQ = \langle value \rangle$ and $IQ = \langle value \rangle$.

IFAIL = 11

On entry, LR is not large enough, $LR = \langle value \rangle$: minimum length required = $\langle value \rangle$.

IFAIL = 12

On entry, STATE vector has been corrupted or not initialized.

IFAIL = 20

Invalid sequence generated, use different parameters.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.
See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.
See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.
See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05PGF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

None.

10 Example

This example first calls G05KFF to initialize a base generator then calls G05PGF to generate two realizations, each consisting of ten observations, from an exponential GARCH(1,1) model.

10.1 Program Text

```

Program g05pgfe

!      G05PGF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: g05kff, g05pgf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: lseed = 1, nin = 5, nout = 6
!      .. Local Scalars ..
Integer                    :: df, genid, i, ifail, ip, iq, lr,      &
                          lstate, ltheta, nreal, num, rn,          &
                          subid
Logical                    :: fcall
Character (1)              :: dist
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: et(:), ht(:), r(:), theta(:)
Integer                    :: seed(lseed)
Integer, Allocatable        :: state(:)
!      .. Executable Statements ..
Write (nout,*) 'G05PGF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the base generator information and seed
Read (nin,*) genid, subid, seed(1)

!      Initial call to initializer to get size of STATE array
lstate = 0
Allocate (state(lstate))
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Reallocate STATE
Deallocate (state)
Allocate (state(lstate))

!      Initialize the generator to a repeatable sequence
ifail = 0
Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!      Read in sample size and number of realizations
Read (nin,*) num, nreal

!      Read in number of coefficients
Read (nin,*) ip, iq

lr = 2*(ip+2*iq+2)
ltheta = 2*iq + ip + 1
Allocate (theta(ltheta),ht(num),et(num),r(lr))

!      Read in error distribution
Read (nin,*) dist

!      Read in degrees of freedom if required
If (dist=='T' .Or. dist=='t') Then
  Read (nin,*) df
End If

!      Read in rest of series parameters
Read (nin,*) theta(1:ltheta)

!      Set FCALL for first realization
fcall = .True.

```

```

!      Generate NREAL realizations
      Do rn = 1, nreal

          ifail = 0
          Call g05pgf(dist,num,ip,iq,theta,df,ht,et,fcall,r,lr,state,ifail)

!      Display the results
      Write (nout,99998) 'Realization Number ', rn
      Write (nout,*) '      I              HT(I)              ET(I)'
      Write (nout,*) '      -----'
      Write (nout,99999)(i,ht(i),et(i),i=1,num)
      Write (nout,*)

!      Set FCALL flag for any further realizations
      fcall = .False.
      End Do

99999 Format (1X,I5,1X,F16.4,1X,F16.4)
99998 Format (1X,A,I0)
      End Program g05pgfe

```

10.2 Program Data

```

G05PGF Example Program Data
1 1 1762543      :: GENID,SUBID,SEED(1)
10 2           :: NUM,NREAL
1 1           :: IP,IQ
'N'
0.1 -0.3 0.1 0.9  :: THETA

```

10.3 Program Results

G05PGF Example Program Results

```

Realization Number 1
      I              HT(I)              ET(I)
      -----
      1              2.5098              0.5526
      2              2.1785             -1.8383
      3              3.3844              1.2180
      4              2.6780              1.3672
      5              2.0953             -1.8178
      6              3.2813             -0.0343
      7              2.9958             -0.5094
      8              3.0815              1.3978
      9              2.3961             -0.0070
     10              2.2445              0.6661

```

```

Realization Number 2
      I              HT(I)              ET(I)
      -----
      1              1.9327             -2.2795
      2              3.5577             -1.2249
      3              4.1461              0.6424
      4              3.4455             -2.9920
      5              5.9199              0.5777
      6              4.8221             -1.2894
      7              5.3174             -1.6473
      8              6.1095              6.1689
      9              3.1579              2.2935
     10              2.2189              0.1141

```